Gains and pitfalls of quantifier elimination as a teaching tool

Reinhard Oldenburg

Angaben zur Veröffentlichung / Publication details:

Oldenburg, Reinhard. 2015. "Gains and pitfalls of quantifier elimination as a teaching tool." *The International Journal for Technology in Mathematics Education* 22 (4): 163–67. https://doi.org/10.1564/tme_v22.4.04.

Gains and Pitfalls of Quantifier Elimination as a Teaching Tool

By Reinhard Oldenburg

Universität Augsburg, Universitätsstr. 14, 86159 Augsburg, Germany,
reinhard.oldenburg@math.uni-augsburg.de

Received: 2 December 2014 Revised: 24 February 2015 **DOI: 10.1564/tme_v22.4.04**

Quantifier Elimination is a procedure that allows simplification of logical formulas that contain quantifiers. Many mathematical concepts are defined in terms of quantifiers and especially in calculus their use has been identified as an obstacle in the learning process. The automatic deduction provided by quantifier elimination thus allows students to exercise the formulation of concepts using quantifiers. This may be seen as conceptual modelling.

1 INTRODUCTION

Tarski (1948) has shown that formulae of first order predicate logic over certain fields can be decided algorithmically. Nowadays this is based on ideas from algebraic geometry namely the method of algebraic cylindrical decomposition. An overview of the method is given in Artigue & Oldenburg (2014).

Here is a non-trivial example: For what values of p does the polynomial $f(x) = x^2 + px + 9$ have real roots with distance apart of at least 1? In mathematical notation solving this problem is the same as turning the logical formula $\exists x_1 \exists x_2 : x_1 < x_2 + 1 \wedge f(x_1) = 0 \wedge f(x_2) = 0$ into a version that is free of quantifiers. In computer algebra syntax this can be solved as follows:

```
f(x):= x^2+p*x+9$
qe([[E,x1],[E,x2]],
  x1<x2+1 %and f(x1)=0 %and f(x2)=0);
```

Here the capital E stands for the existential quantifier and qe is the function that eliminates quantifiers from a formula. The result of this command is `(p-6>=0) %or (p+6<=0)`. That this problem could be solved is not a special property of this particular problem but Tarski's result guarantees that all quantifier elimination problems in a wide range of expressions can be solved by an algorithm.

Tarski himself noted that this leads to a decision procedure for elementary geometry as well. Furthermore, it gives a systematic way to solve systems of polynomial inequalities over $\mathbb{R}$. Many notions from calculus that are expressed in terms of quantifiers can be formalized and decided for purely algebraic functions. This shows that the method of quantifier elimination is suited for several classes of problems that are relevant in mathematics education at various levels. Thus, the question arises, whether this method can be used as a teaching tool. One may hope that having access to quantifier elimination in a computer algebra system may give students the opportunity to explore the mentioned fields of application. Especially one may hope that this may provide a playground to exercise the formalisation step in mathematics. For example, one may have an intuitive idea of what it means for a function to be convex on an interval but it is a crucial further step to be able to formalize this in the language of predicate calculus. We give examples of all kinds of didactically relevant applications and especially example on the formalizations of notions. Based on this example set we systematize the potential and the inherent problems of quantifier elimination as a teaching method.

This paper furthermore presents a didactical theory that highlights the similarity between modelling and formalization.

2 QUANTIFIER ELIMINATION

Universal and existence quantifiers are the basic tools of predicate logic and they allow expressing many mathematical statements. In general, the truth of statements in unrestricted predicate logic is undecidable. However, Tarski's ingenious contribution is to identify a very strong subset that can indeed be decided. The domain of variables in this theory is the set of real numbers, quantifiers are of first order (i.e. they concern variables for numbers not for functional symbols), propositional logic expressions (conjunction, disjunction, negation, implication), relations <, =, > among numbers and polynomials in the variables. The restriction to polynomials is not severe. In fact, the qepmax implementation in Maxima (Honda, 2014) pre-processes e.g. rational equations by applying the following transformation rules: $\frac{a}{b} = c \Leftrightarrow a = bc \wedge b \neq 0$. Square roots and absolute values can be treated as well: $a = \sqrt{b} \Leftrightarrow a^2 = b \wedge a \geq 0, a = |b| \Leftrightarrow a^2 = b^2 \wedge a \geq 0$but this is not yet implemented in the current version of qepmax.

For this language, Tarski has shown that quantifiers can be eliminated from all such formulae and that thus that it is possible to do algorithmic proving of statements in this language. E.g. eliminating the quantifier from $\forall x(x+1)^2 = x^2 + 2x + 1$gives true. This is trivial, so let's look at a more interesting example: For which values of the parameter c does the function $f(x)=x^3-cx^2+cx+c$ have at least two distinct real roots? The statement to decide is $\exists x_1 : \exists x_2 : x_1 \neq x_2 \wedge f(x_1) = 0 \wedge f(x_2) = 0$.

Eliminating quantifiers from this gives a condition on c, namely $c \leq -1$ or $c \geq 27/5$, that precisely describes all values for which there are at least two roots. The calculation carried out in the computer algebra system Maxima using the qepmax library is shown below in Figure 1.

```
(%i146) f1(x):=x^3-c*x^2+c*x+c;
(%o146) f1(x):=x^3-c x^2+c x+c

(%i147) qe([[E,x1],[E,x2]],x1#x2 %and f1(x1)=0 %and f1(x2)=0);
(%o147) (c+1<=0) %or (5 c-27>=0)
```

Figure 1 Performing quantifier elimination in Maxima

Qepmax extends Maxima syntax by logical operators `%and`, `%or` and `%implies`. Quantifiers are given in a list where E and A indicate existential and universal quantification. The only top level function that users will need is the quantifier elimination function qe. Qepmax is basically an interface to the qepcad program. Y. Honda has implemented the inter-process communication and the author of the present paper has contributed the ability to handle rational expressions, not just polynomial expressions.

Application areas of quantifier elimination are wider than proving algebraic identities as many problems can be translated into algebraic-logical language. This includes proving theorems in elementary geometry and exact solution of (possibly constrained) optimization problems.

As most notions of calculus like limit, continuity, convexity, differentiability are defined in terms of quantifiers, one can use quantifier elimination to make these notions computable at least for a restricted set of functions defined by algebraic expressions.

This paper will investigate the scope of an approach to concepts of calculus based on this method.

3 APPLICATIONS IN CALCULUS

The need to work with quantifiers has been identified as one of the key obstacles in learning calculus (e.g. Roh, 2010 and Durand-Guerrier, 2012). Working with a computer algebra system that supports quantifier elimination allows students to formalize notions and then test if the formal expression really gives the behaviour they intended. The quantifier elimination method thus gives them a quick response that is guaranteed to be formally correct.

We view this whole process as a special case of modelling (Niss, Blum and Galbraith, 2007). Students may have developed e.g. from looking at examples an intuitive notion, a concept image (Tall and Vinner, 1981). Now, this informal notion is formalized to become a sound logical formula. Finally, this is translated into the syntax of the computer algebra system and can then be applied. Interpretations of results may make it necessary to refine the definition so that it meets the concept image that should be modelled – or it may lead to the insight that the informal concept image is not well formed and should be altered. The following diagram (Figure 2) shows a modelling circle (Niss et al. 2007) applied to the concept of global minimum. In this example the informal concept image might first have the property that the minimum is unique, which is, however, not encoded in the formal definition as the example shows. This might either lead to modify the informal or the formal definition or it might lead to introduce two notions.



Figure 2 Modelling circle for modelling concepts within a formal system

We will now give some first examples and to this end define some test functions that span a range of polynomial and rational functions with various types of singularities.

```
(%i2) p1(x):=(x-5)^2+x+1; p2(x):=x^3-x;
(%o2) p1(x):=(x-5)^2+x+1
(%o3) p2(x):=x^3-x

(%i4) r1(x):=x/(x-3); r2(x):=1/((x-2)*(x-1)); r3(x):=-1/(1+x^2);
(%o4) r1(x):=x/(x-3)
(%o5) r2(x):=1/((x-2)(x-1))
(%o6) r3(x):=-1/(1+x^2)
```

Figure 3 Some test functions

The notion that a function has a global maximum can be expressed in the form of two quantifiers, as. Figure 4 illustrates. Note that in Maxima one may define functions on the fly by using the lambda keyword: lambda([parameters],function expression). This comes in handy when you need to pass a function to another function which you don't want to define explicitly.

```
(%i7) hasGlobalMin(f):= qe([[E,x0],[A,x]], f(x0)<=f(x)) ;
(%o7) hasGlobalMin(f):=qe([[E,x0],[A,x]],f(x0)<=f(x))

(%i8) [hasGlobalMin(p1),hasGlobalMin(p2)];
(%o8) [true,false]

(%i9) [hasGlobalMin(r1),hasGlobalMin(r2),hasGlobalMin(r3)];
(%o9) [false,false,true]

(%i10) hasGlobalMin(lambda([x],(x-1)^2*(x+1)^2));
(%o10) true

(%i11) hasUniqueGlobalMin(f):= qe([[E,x0],[A,x]], x#x0 %implies f(x0)<f(x));
(%o11) hasUniqueGlobalMin(f):=qe([[E,x0],[A,x]],x#x0 %implies f(x0)<f(x))

(%i12) hasUniqueGlobalMin(lambda([x],(x-1)^2*(x+1)^2));
(%o12) false
```

Figure 4 Global minima (# in Maxima means 'not equal'. Anonymous functions are defined in Maxima using the lambda function constructor)

Local minima are an even richer field for exploration. The absolute value function that is needed to describe distances is not available in the current implementation of qepcad so that one needs to express $|x| < y$ as $x^2 < y^2 \wedge y \geq 0$ and similar for other uses (probably this may require introducing new variables) – see figure 5.

```
(%i17) hasLocalMin(f):= qe([[E,x0],[E,delta],[A,x]],
         delta>0 %and (((x-x0)^2<delta^2) %implies (f(x0)<=f(x)) )) ;
(%o17) hasLocalMin(f):=qe([[E,x0],[E,δ],[A,x]],δ>0 %and ((x-x0)^2<δ^2 %implies f(x0)<=
(%i18) hasLocalMin(lambda([x], x^2*(x-1) ));
(%o18) true
(%i19) hasLocalMin(lambda([x], x^3 ));
(%o19) false
(%i20) hasLocalMin(lambda([x], -x^2 ));
(%o20) false
```

Figure 5 Formalizing local minima.

Up to now, all variables in the formulae were bound by quantifiers. Thus eliminating them yielded either false or true as result. Variables that are not bound by quantifiers are in a sense much more interesting, as they allow the method to calculate conditions on them to make the statement true as the next example, in Figure 6, shows:

```
(%i21) hasLocalMinIn(f,x0):= qe([[E,delta],[A,x]],
         delta>0 %and (((x-x0)^2<delta^2) %implies (f(x0)<=f(x)) )) ;
(%o21) hasLocalMinIn(f,x0):=qe([[E,δ],[A,x]],δ>0 %and ((x-x0)^2<δ^2 %implies f(x0)<=f(x)))
(%i22) hasLocalMinIn(lambda([x], x^2*(x-1) ),X);
(%o22) 3X-2=0
```

Figure 6 Leaving variables free often produces interesting results!

Next, we consider the notion of continuity. The techniques to translate the standard formalization are similar to the examples above so we hope that the screen shots (figure 7) speak for themselves.

```
(%i54) continous(f,x0):= qe([[A,eps],[E,delta],[A,x]],
         (delta>0) %and (
         (eps>0)  %implies
          (( ((x-x0)^2<delta^2) )
              %implies ((f(x)-f(x0))^2<eps^2) )));
(%o54) continous(f,x0):=qe([[A,eps],[E,δ],[A,x]],δ>0 %and
(eps>0 %implies ((x-x0)^2<δ^2 %implies (f(x)-f(x0))^2<eps^2)))
(%i55) p1(x):=(x-5)^2+x+1$ p2(x):=x^3-x$ r1(x):=x/(x-3)$
       r2(x):=1/((x-2)*(x-1))$ r3(x):=-1/(1+x^2)$
(%i60) continous(p1,2);
(%o60) true
(%i61) continous(r1,2);
(%o61) true
(%i62) continous(r1,a);
(%o62) a-3#0
```

Figure 7 Translating absolute value conditions to squares is a bit clumsy but works as expected. Especially, the method can detect where rational functions are not continuous.

Besides rational functions there are no examples where the question of continuity is interesting in the range of functions that can be applied. However, using a simple trick one can encode piecewise defined functions by logical propositions by means of the equivalence:

$$f(x) = \begin{Bmatrix} f_1(x) if x < x_0 \\ f_2(x) if x \geq x_0 \end{Bmatrix} \wedge y = f(x)$$
$$\Leftrightarrow \left(x < x_0 \Rightarrow y = f_1(x)\right)$$
$$\wedge \left(x \geq x_0 \Rightarrow y = f_2(x)\right)$$

This requires rewriting functions in a relational way. The square function which is in Maxima defined by `f(x):=x^2` in functional fashion will be replaced by `fR(x,y):= y=x^2`. The conversion can be done automatically:
`convertToR(f):= apply(lambda,[[x,y], y=f(x)])` so that e.g. `sqrtR:convertToR(f)` gives a relational sqrt function. With this bag of tricks piecewise defined functions can be handled. The function `makePWfunglg` takes two functions of this relational kind and a number x0 and produces a new function for which the first function defines the interval up to x0 and the second the interval from this point on.

```
(%i45) makePWfunglg(f1,x0,f2):=
         apply(lambda,[[x,y],
        ((x<x0) %implies (y=f1(x))) %and ((x>=x0) %implies (y=f2(x)))]);
(%o45) makePWfunglg(f1,x0,f2):=apply
(lambda,[[x,y],(x<x0 %implies y=f1(x)) %and (x>=x0 %implies y=f2(x))])
(%i46) makePWfunglg(lambda([x],x),0,lambda([x],x^2))(5,y);
(%o46) (5>=0 %implies y=25) %and (5<0 %implies y=5)
(%i47) continousFG(fg,x0):= qe([[E,y0],[A,eps],[E,delta],[A,x],[E,y]],
          (delta>0) %and ((eps>0) %implies
                 (( ((x-x0)^2<delta^2) ) %implies
                  ( fg(x,y)  %and fg(x0,y0) %and  (y-y0)^2<eps^2) )));
(%o47) continousFG(fg,x0):=qe([[E,y0],[A,eps],[E,δ],[A,x],[E,y]],δ
>0 %and (eps>0 %implies
((x-x0)^2<δ^2 %implies fg(x,y) %and fg(x0,y0) %and (y-y0)^2<eps^2)))
(%i48) continousFG(makePWfunglg(lambda([x],x),0,lambda([x],x^2+x)),0);
(%o48) true
(%i49) continousFG(makePWfunglg(lambda([x],x+1),0,lambda([x],x^2+x)),0);
(%o49) false
```

Figure 8: Piecewise defined functions and the test for continuity

The same considerations apply for testing differentiability. However, it turns out that proving the differentiability of piecewise defined functions would take too much memory to be handled by Maxima on standard computers and may be out of reach as the complexity of the quantifier elimination algorithm can be very time consuming (e.g. double exponential in the number of variables). However, the implementation of quantifier elimination in Mathematica (Wolfram 2015) can handle this example.

```
(%i50) DiffGlg(fg,x0,m):=qe([[E,y0],[A,eps],[E,delta],[A,x],[E,y]],
          (eps>0 %and x#x0) %implies
              (delta>0 %and (((x-x0)^2<delta^2) %implies
               ( fg(x,y) %and fg(x0,y0) %and (m-(y-y0)/(x-x0))^2<eps^2)
(%o50) DiffGlg(fg,x0,m):=qe([[E,y0],[A,eps],[E,δ],[A,x],[E,y]],
eps>0 %and x#x0 %implies δ>0 %and
((x-x0)^2<δ^2 %implies fg(x,y) %and fg(x0,y0) %and (m-(y-y0)/(x-x0))^2<eps^2))
(%i51) DiffGlg(lambda([x,y],y=x^2),2,d);
(%o51) d-4=0
(%i52) DiffGlg(makePWfunglg(lambda([x],x),0,lambda([x],x^2)),2,d);
bash: Zeile 1:  8835 Abgebrochen             (Speicherabzug geschrieben)
(%o52) FAIL
```

Figure 9 Maxima proving that the derivative of $f(x) = x^2$ at 2 is 4. It fails however, to establish differentiability of a piecewise defined function.

```
In[1]:= implies[a_, b_] := Not[a] || b

In[2]:= diff[fg_, x0_, d_] :=
         Exists[y0, fg[x0, y0] && ForAll[eps, implies[eps > 0, Exists[delta,
               delta > 0 && ForAll[x, Exists[y, implies[x ≠ x0, fg[x, y] && implies[
                     Abs[x - x0] < delta, Abs[(y - y0) / (x - x0) - d] < eps]]]]]]]

In[3]:= quadrat[x_, y_] := y == x^2

In[4]:= diff[quadrat, 3, d]

Out[4]= ∃y0 (y0 == 9 && ∀eps (eps ≤ 0 || ∃delta (delta > 0 &&
          ∀x ∃y (x == 3 || (y == x^2 && (Abs[-3 + x] ≥ delta || Abs[-d + (-y0 + y)/(-3 + x)] < eps))))))

In[5]:= Resolve[diff[quadrat, 3, d], Reals]

Out[5]= d == 6

In[6]:= makePW[f1_, x0_, f2_] :=
         Function[{x, y}, implies[x < x0, y == f1[x]] && implies[x >= x0, y == f2[x]]]

In[7]:= Resolve[diff[makePW[Function[{x}, x], 0, Function[{x}, x^2 + x]], 0, d], Reals]

Out[7]= d == 1

In[8]:= Resolve[diff[makePW[Function[{x}, x], 0, Function[{x}, x^2 + x]], 1, d], Reals]

Out[8]= d == 3

In[9]:= Resolve[diff[makePW[Function[{x}, 2 * x], 0, Function[{x}, x^2 + x]], 0, d], Reals]

Out[9]= False
```

Figure 10 Mathematica can determine the derivative (see cell In[8]) for which Maxima (Figure 9) failed

Next, we consider two notions that are linked to calculus without involving limits: monotone and convex functions.

```
(%i29) monotoneDe(f,a,b):= qe([[A,x1],[A,x2]],
         (a<x1 %and x1<x2 %and x2<b) %implies f(x2)<f(x1));
(%o29) monotoneDe(f,a,b):=
qe([[A,x1],[A,x2]],a<x1 %and x1<x2 %and x2<b %implies f(x2)<f(x1))

(%i30) monotoneIn(f,a,b):= qe([[A,x1],[A,x2]],
         (a<x1 %and x1<x2 %and x2<b) %implies f(x2)>f(x1));
(%o30) monotoneIn(f,a,b):=
qe([[A,x1],[A,x2]],a<x1 %and x1<x2 %and x2<b %implies f(x2)>f(x1))

(%i31) monotoneIn(lambda([x],x*(x-1)*(x-2)),a,b);
(%o31) (a-1>0) %and (3 a^2-6 a+2>=0) %or (b-a<=0) %or
((b-1<0) %and (3 b^2-6 b+2>=0))
```

Figure 11 Functions that are strictly monotone in an interval. Obvious variations are non-strict versions.

4 THE LIMITATIONS

The last section has shown some glimpses of the power of the method of quantifier elimination. The most important is that it is a correct method so that students get results they can trust (unless the computing time or memory usage becomes too high to be acceptable). This is a crucial point. In Oldenburg and Weygandt (2015) we showed how much wrong or incomplete answers from computing limits within a computer algebra system can irritate students. However, the correctness of quantifier elimination comes at a price: the set of test functions is restricted to the algebraic class described above, so that all transcendental functions like exponentials, logarithms and trigonometric functions as well as special functions that are of interests e.g. for physicists (Airy, Bessel, ...) are out of the reach of the method. Piecewise defined functions have to be handled in way that is not very user friendly (although this drawback might be overcome by smoother interfaces to the core method). In its direct incarnation of the method not even rational functions can be handled which leads the students with a set of functions that includes only continuous and differentiable ones. The Maxima implementation is thus an important step from a didactical perspective.

```
(%i32) convexity(f,a,b):= qe([[A,x1],[A,x2],[A,x3]],
         a<b %and ((a<x1 %and x1<x2 %and x2<x3 %and x3<b)
              %implies f(x2)<f(x1)+(x2-x1)*(f(x3)-f(x1))/(x3-x1)));
(%o32) convexity(f,a,b):=qe([[A,x1],[A,x2],[A,x3]],a<b %and (a<x1
%and x1<x2 %and x2<x3 %and x3<b %implies f(x2)<f(x1)+
(x2-x1)(f(x3)-f(x1))/(x3-x1)))

(%i33) convexity(lambda([x],x^2),1,5);
(%o33) true

(%i34) convexity(lambda([x],x*(x-1)*(x-2)),1,a);
(%o34) a-1>0
```

Figure 12 Convexity detection without using second derivative

A question that pushes the complexity of logical combinations to the limit of the system (and maybe the user) is to express that a point is an inflection point of a function in the sense that the function is convex on one side and concave on the other.

```
(%i39) inflectionpoint(f,x3,a,b):=
       qe([[A,x1],[A,x2],[A,x4],[A,x5]],
       a<x3 %and x3<b %and
       (a<x1 %and x1<x2 %and x2<x3 %and x2<x3 %and x3<x4
        %and x4<x5 %and x5<b)
          %implies
          (       (f(x2)<f(x1)+(x2-x1)*(f(x3)-f(x1))/(x3-x1))
            %and (f(x4)>f(x3)+(x4-x3)*(f(x5)-f(x3))/(x5-x3))
           %or     (f(x2)>f(x1)+(x2-x1)*(f(x3)-f(x1))/(x3-x1))
             %and (f(x4)<f(x3)+(x4-x3)*(f(x5)-f(x3))/(x5-x3)))
       );
(%o39) inflectionpoint(f,x3,a,b):=qe([[A,x1],[A,x2],[A,x4],[A,x5]]
,a<x3 %and x3<b %and (a<x1 %and x1<x2 %and x2<x3 %and x2<x3 %and x3
<x4 %and x4<x5 %and x5<b) %implies (f(x2)<f(x1)+(x2-x1)(f(x3)-f(x1))/(x3-x1)
%and f(x4)>f(x3)+(x4-x3)(f(x5)-f(x3))/(x5-x3) %or f(x2)>f(x1)+(x2-x1)(f(x3)-f(x1))/(x3-x1)
%and f(x4)<f(x3)+(x4-x3)(f(x5)-f(x3))/(x5-x3)))

(%i40) inflectionpoint(lambda([x],x*(x-1)*(x-2)),x,-5,5);
(%o40) (x-5>=0) %or (x-1=0) %or (x+5<=0)
```

Figure 13 Checking if there is an inflection point of $f(x) = x(x-1)(x-2)$ in the interval [-5,5]. Note that if is not from this interval no statement is made, so it is not false!

Another drawback is the vast computing power to carry out the methods – and that it increases very fast with the complexity of the problem. One may say that from the perspective of a trained mathematician the method is only up to toy problems.

5 BALANCING GAINS AND PITFALLS

Often the claim is made that the most advanced mathematics and most sophisticated implementation is not needed for education as students deal with elementary concepts. Quantifier elimination is certainly an exception to this. Only now the implementations get so fast and reliable that the method can provide a safe and reliable 'playground' for working with quantifiers. And this advanced mathematics is very tightly linked to basic mathematical concepts like systems of inequalities over real numbers. One may imagine that future educational applications may include learning environments that give students the opportunity to model situations by means of equations and inequations and draw conclusions that can be checked by the computer.

A further didactical perspective that opens up is to treat logic earlier in the curriculum. For example, one may look at the inequality between the geometric mean and the arithmetic mean both from algebraic and from geometric perspectives and may elaborate the case distinction between strict and non-strict versions of the inequality.

But most likely, the method will proliferate from university math downwards with calculus being an interesting field of application as we have shown in this paper. We hope to incorporate quantifier elimination based exercises in a future calculus course. A crucial question will be what kind of formalization errors students commit and how they can be addressed. Furthermore, one should study if students are able to come up with interesting test functions linking this to the subject of learner generated examples.

We expect the method to become in more widespread use in the next years as it gains momentum from several directions including exact optimization and automated theorem proving in geometry. This will hopefully increase the use in education as well and will provide empirical evidence to give a judgement if he gains outweigh the pitfalls.

REFERENCES

Artigue, M. and Oldenburg, R. (2014) How to get rid of quantifiers *Klein Vignette*, Available at: http://blog.kleinproject.org/?p=2466

Durand-Guerrier, V. (2012) An insight on university mathematics teaching practices about proofs involving multiple quantifiers. Available at: http://www.icme12.org/upload/UpFile2/POSTERS/1615.pdf.

Honda, Y. : Qepmax (2014) Maxima-qepcad-interface https://github.com/YasuhaikHonda/qepmax

Niss, M., Blum, W. and Galbraith, P. (2007) Introduction, in Blum, W., Galbraith, P., Henn, H-W. and Niss, M. (eds) *Modelling and applications in mathematics education – The 14th ICMI study*, New-York, USA: Springer, 4-32.

Oldenburg, R. and Weygandt, B. (2015): Einsatzmöglichkeiten und Grenzen von Computeralgebrasystemen zur Förderung des Begriffsverständnisses, *Proceedings of khdm 2013*, in press.

Tarski, A. (1948) *A decision method for elementary algebra and geometry*, Rand Corporation Publication. Republished as A Decision Method for Elementary Algebra and Geometry, 2nd ed. Berkeley, CA: University of California Press, 1951.

Roh, K. H. (2010) An empirical study of students' understanding of a logical structure in the definition of limit via the ε-strip activity, *Educational Studies in Mathematics*, **73**, 263–279

Tall, D., and Vinner, S. (1981) Concept image and concept definition in mathematics with particular reference to limits and continuity, *Educational Studies in Mathematics,* **12**, 151-169.

Wolfram, S. (2015): Mathematica – A system for doing mathematics by Computer. http://www.wolfram.com/mathematica/

BIOGRAPHICAL NOTES

Reinhard Oldenburg studied mathematics, physics and computer science and took a PhD in mathematics. He has been a professor of mathematics education in Heidelberg and Frankfurt and currently at Augsburg university.